

**UNIVERSIDADE FEDERAL DE MINAS GERAIS
FACULDADE DE FILOSOFIA E CIÊNCIAS HUMANAS
DEPARTAMENTO DE CIÊNCIA POLÍTICA**

Curso de Introdução ao Stata – Aula 2

Professor

Ernesto Friedrich de Lima Amaral

Email: eflamaral@gmail.com

Site do curso: www.ernestoamaral.com/stata2012.html

Inclusão de observações de outros bancos (*append*)

Vamos utilizar o comando "append" (inclusão de observações) para juntar os bancos de Minas Gerais e Goiás:

```
use "C:\curso\dados\pes2007MG.dta", clear
append using "C:\curso\dados\pes2007GO.dta", nolabel
save "C:\curso\dados\pes2007MGGGO.dta", replace
```

Acima foi utilizada a opção "nolabel" para excluir os rótulos das categorias do segundo banco. Isso é importante de ser realizado quando determinadas variáveis possuem rótulos diferentes entre os bancos que estão sendo agrupados.

Inclusão de variáveis de outros bancos (*merge*)

Suponha que queremos incluir as variáveis de domicílio de Minas Gerais no banco de indivíduos. Para isto, utilizamos o comando "merge" (inclusão de variáveis).

Antes de tudo, é preciso ordenar os bancos pelas variáveis de identificação. No caso da PNAD, as variáveis de identificação são número de controle (v0102) e número de série (v0103).

No entanto, incluímos também ano de referência (v0101) e Unidade da Federação (uf), simplesmente porque elas aparecem nos dois bancos. Se estas duas variáveis não fossem incluídas no comando, o resultado acabaria sendo o mesmo, já que os dados do domicílio seriam gravados sobre os dados de indivíduos.

Primeiramente, ordenamos o banco de domicílios:

```
use "C:\curso\dados\dom2007MG.dta", clear
sort v0101 uf v0102 v0103
save "C:\curso\dados\dom2007MG.dta", replace
```

Em seguida, ordenamos o banco de indivíduos e realizamos a junção dos dados:

```
use "C:\curso\dados\pes2007MG.dta", clear
sort v0101 uf v0102 v0103
merge v0101 uf v0102 v0103 using "C:\curso\dados\dom2007MG.dta"
```

Note que a variável "_merge" é criada automaticamente pelo Stata e pode assumir os seguintes resultados:

_merge=1 observações somente do banco de dados mestre (*master data*)
 _merge=2 observações somente do banco de dados secundário (*using data*)
 _merge=3 observações dos dois bancos (*master e using*)

O ideal é que todas as observações de "_merge" tenham valor igual 3, já que isso seria um indício de que houve a junção de dados de domicílios e pessoas em todos os casos.

No entanto, observamos que há 2.906 casos que são oriundos somente do banco de domicílios:

```
tab _merge
```

_merge	Freq.	Percent	Cum.
2	2,906	7.41	7.41
3	36,320	92.59	100.00
Total	39,226	100.00	

É preciso investigar porque isso aconteceu. Vamos olhar todas variáveis para os casos em que "_merge" é igual a 2:

```
browse if _merge==2
```

Observando o banco, vemos que a primeira variável de domicílio (tipo de entrevista – v0104) possui valores diferentes de "missing", mas as variáveis seguintes apresentam valores em branco. Vamos ver os valores dessa variável quando "_merge" é igual a 3:

```
tab v0104 if _merge==3, missing
```

Os 36.320 casos indicam que a entrevista foi realizada (v0104=1).

Agora vamos ver a tabela desta variável "v0104" somente no caso de "_merge" ser igual a 2:

```
tab v0104 if _merge==2, missing
```

Perceba que neste segundo caso, as observações são de entrevistas não realizadas (v0104 diferente de 1):

TIPO A UNIDADE OCUPADA

v0104=02 (Fechada)

v0104=03 (Recusa)

v0104=04 (Outra)

TIPO B UNIDADE VAGA

v0104=05 (Em condições de ser habitada)

v0104=06 (Uso ocasional)

v0104=07 (Construção ou reforma)

v0104=08 (Em ruínas)

TIPO C UNIDADE INEXISTENTE

v0104=09 (Demolida)

v0104=10 (Não foi encontrada)

v0104=11 (Não residencial)

v0104=12 (Fora do setor)

Ou seja, o total de observações em que as entrevistas não foram realizadas é o mesmo que o número de vezes em que "_merge" é igual a dois.

Tipo de entrevista	Freq.	Percent	Cum.
2	286	9.84	9.84
3	112	3.85	13.70
4	16	0.55	14.25
5	1,182	40.67	54.92
6	862	29.66	84.58
7	202	6.95	91.53
8	67	2.31	93.84
9	73	2.51	96.35
10	30	1.03	97.38
11	76	2.62	100.00
Total	2,906	100.00	

No decorrer do banco, algumas variáveis de domicílio apresentam valores diferentes de "missing", a partir da variável "v4105". Estas são variáveis derivadas que são possíveis de ser captadas mesmo não havendo realização da entrevista.

Vimos que não houve problema na realização do "merge" entre bancos de domicílios e pessoas. Por isso, podemos excluir os casos em que "_merge" é igual a dois e salvar o banco:

```
drop if _merge==2
drop _merge
save "C:\curso\dados\pesdom2007MG.dta", replace
```

Gráficos

De uma forma geral, é bom olhar o menu *Graphics* para explorar os diversos tipos de gráficos elaborados pelo Stata. Aqui vou colocar alguns exemplos:

– Gráfico de barras:

Média do rendimento mensal no trabalho principal (v4718), de todos trabalhos (v4719) e de todas as fontes (v4720) por raça (branca X não branca).

Recodificando variável cor/raça:

```
gen branca=.
replace branca=0 if v0404==4 | v0404==8
replace branca=1 if v0404==2
```

Transformando não declaração de rendimento (999.999.999.999) em "missing":

```
replace v4718=. if v4718==999999999999
replace v4719=. if v4719==999999999999
replace v4720=. if v4720==999999999999
```

Definindo o rótulo da variável cor/raça:

```
label define branca 0 "Preta/Parda" 1 "Branca"
label values branca branca
```

Gerando o gráfico de barras:

```
graph bar (mean) v4718 v4719 v4720, over(branca) ///
///
title("Média de rendimento de trabalho principal, de todos trabalhos," ///
"de todas as fontes por cor/raça em Minas Gerais, PNAD 2007" ///
, size(medlarge)) ///
///
ytitle("Média de rendimento") ///
///
xlabel(bar, format(%9,2fc)) ///
///
bar(1, fcolor(dknavy) lcolor(dknavy)) ///
bar(2, fcolor(gray) lcolor(gray)) ///
bar(3, fcolor(dkgreen) lcolor(dkgreen)) ///
///
legend(title(Tipos de rendimento, size(medsmall)) ///
label(1 "Trabalho principal") ///
label(2 "Todos trabalhos") ///
label(3 "Todas as fontes"))
```

Salvando o gráfico como figura:

```
graph export "C:\curso\grafs\medrenda.wmf", replace
```

– Histogramas:

Transformando anos de estudo não-determinados (v4803=17) em "missing":

```
gen estudo=.
replace estudo=v4803-1 if (v4803>=1 & v4803<=16)
```

Histograma de anos de estudo:

```
histogram estudo, discrete frequency ///
color(black) lcolor(gray) ///
title("Histograma de anos de estudo em Minas Gerais, PNAD 2007", ///
color(black)) ///
ytitle(Frequência) ///
ylabel(0(2000)8000, grid glcolor(black) gpattern(dash)) ///
xtitle(Anos de estudo) ///
xlabel(1(1)17 17 " ") ///
mcolor(black) fcolor(black) lcolor(white) ///
plotregion(lcolor(black)) ///
graphregion(color(white) icolor(white))
```

Salvando histograma como figura:

```
graph export "C:\curso\grafs\histestudo.wmf", replace
```

Histograma de rendimento do trabalho principal (v4718):

```
histogram v4718, frequency color(black) ///
ytitle(Frequência) ///
xtitle(Rendimento no trabalho principal) ///
title("Histograma de rendimento no trabalho principal" ///
"em Minas Gerais, PNAD 2007")
```

Salvando histograma como figura:

```
graph export "C:\curso\grafs\histv4718.wmf", replace
```

Histograma do logaritmo de rendimento no trabalho principal:

```
gen lnv4718=ln(v4718)

histogram lnv4718, frequency color(black) ///
ytitle(Frequência) ///
ylabel(0(1000)4000, format(%9,0fc)) ///
xlabel(2(1)10) ///
xtitle(Log do rendimento no trabalho principal) ///
title("Histograma do logaritmo de rendimento no trabalho principal" ///
"em Minas Gerais, PNAD 2007")
```

Salvando histograma como figura:

```
graph export "C:\curso\grafs\histlnv4718.wmf", replace
```

– Gráfico de caixa (box-plot):

Distribuição de anos de estudo:

```
label variable estudo "Anos de estudo"

graph hbox estudo, ylabel(0(1)16) ///
title("Distribuição de anos de estudo em Minas Gerais, 2007" ///
, size(large)) ///
note(Fonte: PNAD 2007)
```

Salvando gráfico de caixa de anos de estudo:

```
graph export "C:\curso\grafs\boxestudo.wmf", replace
```

– Gráfico de dispersão (scatterplot):

Idade do indivíduo (v8005) pelo logaritmo do rendimento no trabalho principal (lnv4718):

```
twoway (scatter lnv4718 v8005) (lfit lnv4718 v8005, lwidth(vthick)), ///
legend(label(1 "Logaritmo do rendimento") label(2 "Tendência Linear")) ///
title("Gráfico de dispersão da idade do indivíduo" ///
"pelo logaritmo do rendimento no trabalho principal", ///
color(black) size(large)) ///
xtitle("Idade") ///
xlabel(0(20)100) ///
ytitle("Logaritmo do rendimento") ///
ylabel(1(1)11, grid glcolor(gray) glpattern(dash)) ///
plotregion(lcolor(black)) ///
graphregion(color(white) icolor(white))
```

Salvando gráfico de dispersão:

```
graph export "C:\curso\grafs\idadeXtrab.wmf", replace
```

Utilização do comando foreach para recodificação e criação de grupos de variáveis

As seguintes variáveis possuem o código 999.999.999.999 quando não houve declaração de renda:

- Rendimento mensal no trabalho principal (v4718)
- Rendimento mensal de todos trabalhos (v4719)
- Rendimento mensal de todas as fontes (v4720)
- Rendimento mensal domiciliar (v4721)
- Rendimento mensal familiar (v4722)

Abra o banco de dados original, antes de rodar os comandos abaixo:

```
use "C:\curso\dados\pes2007MG.dta", clear
```

Com o intuito de reclassificar as respostas "sem declaração" para "missing", poderíamos usar todas essas linhas de comando:

```
replace v4718=. if v4718==999999999999
replace v4719=. if v4719==999999999999
replace v4720=. if v4720==999999999999
replace v4721=. if v4721==999999999999
replace v4722=. if v4722==999999999999
```

No entanto, podemos simplesmente escrever:

```
foreach x of varlist v4718-v4722 {
  replace `x'=. if `x'==999999999999
}
```

Depois de recodificar os valores "sem declaração" das variáveis v4718 a v4722, podemos gerar novas variáveis para informar as médias de renda por sexo (v0302):

```
sort v0302
foreach x of varlist v4718-v4722 {
  by v0302: egen med`x' = mean(`x')
}

browse v0302 v4718-v4722 med*
browse v0302 v4718-v4722 med*, nolabel
```

Desagregação de dia, mês e ano de uma variável de data (*nsplit*)

O comando "nsplit.ado" foi retirado deste site (<http://fmwww.bc.edu/repec/bocode/n/nsplit.ado>). Copie o conteúdo dessa página, cole em uma janela do "Do-File Editor" do Stata e salve como "nsplit.ado".

A instalação do módulo "nsplit" pode ser realizada de diferentes maneiras:

1) Copiar manualmente o arquivo "nsplit.ado" para o subdiretório "ado\updates\n" do Stata. Esta é uma maneira de instalar permanentemente o comando no computador em que se está trabalhando.

2) Se não for possível salvar arquivos no computador, mude o diretório de trabalho para o local onde se encontra o arquivo "nsplit.ado":

```
cd "C:\curso\modulos"
```

3) Também é possível rodar o arquivo "nsplit.ado":

```
do "C:\curso\modulos\nsplit.ado"
```

Depois disso, você precisa usar esse comando e dizer para o Stata criar três variáveis ("dia", "mes" e "ano") com dois dígitos cada uma. Para isto, rode o comando abaixo. Neste exemplo, a variável atual do banco de dados é chamada de "data":

```
nsplit data, digits(2 2 2) generate(dia mes ano)
```

Transpor o banco de dados (*xpose*)

Este comando permite transformar variáveis em observações e, como consequência, observações em variáveis:

```
xpose, clear
```

Criação de bancos de dados agrupados por categorias de variáveis (*collapse*)

Vamos criar e salvar um banco de dados que vai indicar médias de rendimentos (v4718-v4722) por sexo (v0302) e cor/raça (branca). Abra o banco de dados original, antes de rodar os comandos abaixo:

```
use "C:\curso\dados\pes2007MG.dta", clear
```

Criando variável cor ou raça: branca X preta/parda:

```
gen branca=.
replace branca=0 if v0404==4 | v0404==8
replace branca=1 if v0404==2
```

Recodificando rendimento:

```
foreach x of varlist v4718-v4722 {
  replace `x'=. if `x'==999999999999
}
```

Calculando médias de rendimentos (v4718-v4722), além do número de observações não-missings e da mediana do rendimento do trabalho principal (v4718), pelas categorias de sexo (v0302) e de cor/raça (branca):

```
sort v0302 branca
collapse (mean) v4718-v4722 (count) c4718=v4718 (median) med4718=v4718, ///
by(v0302 branca)
browse
```

Salvando os dados agrupados:

```
save "C:\curso\dados\collapse.dta", replace
```

Reorganização do arranjo (formato) dos bancos de dados

O comando "reshape" muda o formato dos bancos de dados, segundo a necessidade do estudo em questão. Este é o exemplo da ajuda do Stata:

```
(wide form)
i          ..... x_ij .....
id sex   inc80   inc81   inc82
-----
1    0    5000    5500    6000
2    1    2000    2200    3300
```

```
(long form)
i   j      x_ij
id year sex   inc
-----
1   80    0  5000
1   81    0  5500
1   82    0  6000
2   80    1  2000
2   81    1  2200
2   82    1  3300
```

O comando para mudar o formato dos bancos de "wide" (amplo) para "long" (dados em painel) é:
`reshape long inc, i(id) j(year)`

O comando para mudar o formato dos bancos de "long" (dados em painel) para "wide" (amplo) é:
`reshape wide inc, i(id) j(year)`

Primeiramente, vamos limpar os dados de Minas Gerais, provenientes da PNAD de 2006:

```
use "C:\curso\dados\pes2006MG.dta", clear
keep v0101 uf v0302 v0404 v4718-v4720
save "C:\curso\dados\pes2006MGpeq.dta", replace
```

Agora vamos limpar os dados de Minas Gerais, provenientes da PNAD de 2007 e agrupar com os dados de 2006:

```
use "C:\curso\dados\pes2007MG.dta", clear
keep v0101 uf v0302 v0404 v4718-v4720
append using "C:\curso\dados\pes2006MGpeq.dta"
save "C:\curso\dados\peq2006-2007MG.dta", replace
```

Organizando variável sexo (v0302):

```
gen mulher=.
replace mulher=1 if v0302==4
replace mulher=0 if v0302==2
label define mulher 0 "Homem" 1 "Mulher"
label values mulher mulher
```

Organizando variável cor/raça (v0404):

```
gen branca=.
replace branca=0 if v0404==4 | v0404==8
replace branca=1 if v0404==2
label variable branca "Cor ou raça: branca X preta/parda"
label define branca 0 "Preta/Parda" 1 "Branca"
label values branca branca
```


Organizando variáveis de rendimento (v4718, v4719 e v4720):

```
*Rendimento mensal no trabalho principal (v4718)
foreach x of varlist v4718-v4720 {
  replace `x'=. if `x'==999999999999
  gen ln`x'=ln(`x')
}
```

Salvando dados recodificados:

```
save "C:\curso\dados\peq2006-2007MGrec.dta", replace
```

Agora vamos utilizar o comando "collapse" para calcular as médias de rendimentos (v4718-v4720) por ano, sexo e cor/raça:

```
sort v0101 mulher branca
collapse (mean) v4718-v4720 lnv4718-lnv4720, by(v0101 mulher branca)
```

Salvando os dados agrupados no formato "long":

```
save "C:\curso\dados\collapselong.dta", replace
```

Vamos organizar o ano de referência (v0101):

```
gen ano=.
replace ano=0 if v0101==2006
replace ano=1 if v0101==2007
```

Neste momento, temos os dados em painel (long). Desta forma, podemos estimar modelos de regressão que calculam o impacto do sexo (mulher), cor/raça (branca) em cada um dos rendimentos (v4718, v4719 e v4720), ao longo do tempo (ano). Podemos estimar um modelo da seguinte forma:

```
reg lnv4718 mulher branca ano
```

Podemos passar este banco de dados para o formato amplo (wide). Lembremos o exemplo do Stata:

```
(wide form)
i          ..... x_ij .....
id sex   inc80   inc81   inc82
-----
1    0    5000    5500    6000
2    1    2000    2200    3300
```

```
(long form)
i   j      x_ij
id year sex   inc
-----
1   80    0  5000
1   81    0  5500
1   82    0  6000
2   80    1  2000
2   81    1  2200
2   82    1  3300
```

Antes disso, vamos apagar a variável "v0101", já que ela foi recodificada para "ano":

```
drop v0101
```

O comando para mudar o formato dos bancos de "long" (dados em painel) para "wide" (amplo) é:

```
reshape wide v4718-v4720 lnv4718-lnv4720, ///
i(mulher branca) j(ano)
```

Neste novo formato, podemos estimar correlações entre as variáveis de rendimento nos dois anos em análise.

Alocação de informações entre registros

Vamos supor que queremos colocar algumas respostas da pessoa de referência do domicílio ($v0401=1$) no registro dos outros residentes no mesmo domicílio ($2 \leq v0401 \leq 8$). A idéia é de que essa pessoa de referência teria relação de influência com os demais habitantes no domicílio, principalmente com os filhos ($v0401=3$).

Esse tipo de exercício é muito utilizado em estudos que alocam informações dos pais (renda, educação e outros) para as crianças no mesmo domicílio.

Primeiramente, vamos criar uma variável que identifica a pessoa de referência na unidade domiciliar ($v0401=1$):

```
gen chefe=1 if v0401==1
```

Agora vamos criar as variáveis do chefe do domicílio:

```
foreach x of varlist v0301 v0302 v8005 v0401 v0404 v4803 v4718-v4720 {
  gen c`x' = `x' if chefe==1
}
```

Finalmente, é realizada a alocação das informações do chefe para os demais membros do domicílio:

```
sort v0102 v0103

foreach x of varlist cv0301 cv0302 cv8005 cv0401 cv0404 cv4803 cv4718-cv4720 {
  by v0102 v0103: egen a`x' = max(`x')
}
```

Salvando o banco com dados alocados:

```
save "C:\curso\dados\pes2007MGaloc.dta", replace
```

Teste de Qui-quadrado (*Chi-square test*)

Abra o banco de dados original, antes de rodar os comandos abaixo:

```
use "C:\curso\dados\pes2007MG.dta", clear
```

Antes de realizar o teste de qui-quadrado, é necessário recodificar as variáveis de interesse para colocar os valores "não determinados" e "sem declaração" como "missing":

```
*Grupos de anos de estudo (v4838)
replace v4838=. if v4838==7

*Faixa de rendimento domiciliar per capita (v4743)
replace v4743=. if v4743==99
```

O teste de qui-quadrado é usado quando se deseja saber se há relação entre duas variáveis categóricas. A opção "chi" ou "chi2" é usada com o comando "tabulate" para obter o teste estatístico e seu valor "p".

Vamos testar se há correlação estatisticamente significativa entre grupo de anos de estudo (v4838) e a faixa de rendimento domiciliar per capita (v4743):

```
tabulate v4838 v4743, chi
tabulate v4838 v4743, chi2
```

Observamos que há correlação significativa entre grupo de anos de estudo e faixa de rendimento.

Teste de "t" (*t-test*)

O teste de "t" (*t-test*) permite verificar se a média de uma variável com distribuição normal difere significativamente de um valor hipotético.

Recodificando a variável de anos de estudo:

```
gen estudo=.
replace estudo=v4803-1 if v4803>=1 & v4803<=16
```

Mesmo sabendo que essa variável não tem uma distribuição normal, vamos testar se a média de anos de estudo da população difere significativamente de 7,5, considerando uma escala que varia de 0 a 15:

```
histogram estudo

ttest estudo=7.5
```

A média da variável "estudo" é 5,895, com probabilidade estatisticamente significativa de ser diferente de 7,5 e menor que 7,5.

Podemos verificar se o rendimento mensal do trabalho principal para pessoas de 10 anos ou mais de idade (v4718) tem médias diferentes entre as duas categorias de sexo, com significância estatística:

```
use "C:\curso\dados\pes2007MG.dta", clear

gen mulher=.
replace mulher=1 if v0302==4
replace mulher=0 if v0302==2

replace v4718=. if v4718==9999999999999999

ttest v4718, by(mulher)
```

Como esta variável não tem distribuição normal, podemos calcular o logaritmo da renda, antes de realizar o teste estatístico:

```
gen lnv4718 = ln(v4718)
ttest lnv4718, by(mulher)
```

Intervalo de confiança para variáveis

Estime o intervalo de confiança com nível de confiança de 95% para variável binomial:

```
ci mulher, binomial level(95)
```

Estime o intervalo de confiança com nível de confiança de 95% para variável contínua:

```
ci v4718, level(95)
ci lnv4718, level(95)
```

Correlação

O comando para estimação do coeficiente de correlação do produto de momentos de Pearson é útil para testar se há relação linear entre duas ou mais variáveis com intervalos normalmente distribuídos.

O coeficiente de correlação linear (r) mede a intensidade de relação linear, não sendo planejado para medir intensidade de relação que não seja linear. O valor de r^2 é a proporção da variação em y que é explicada pela relação linear entre x e y .

Podemos usar os comandos "correlate" ou "pwcorr", em que ambos mostram a matriz de correlações entre as variáveis.

O comando "corr" usa "listwise deletion", em que toda matriz é calculada somente para casos que não possuem nenhum valor em branco ("missing") em nenhuma variável na lista:

```
corr x y z
```

O comando "pwcorr" usa "pairwise deletion", em que cada correlação é computada para casos que não possuem nenhum valor em branco para cada par de variáveis:

```
pwcorr x y z, sig
```

Uso do "pwcorr" para obter o mesmo que "corr":

```
pwcorr x y z if !missing(x, y, z), sig
```

Podemos testar se anos de estudo (v4803) estão correlacionados com logaritmo do rendimento mensal no trabalho principal (v4718):

```
use "C:\curso\dados\pes2007MG.dta", clear
gen estudo=.
replace estudo=v4803-1 if v4803>=1 & v4803<=16
gen lnv4718 = ln(v4718) if v4718!=999999999999
```

Em seguida, é realizado o teste de correlação, em que é também possível estimar a significância:

```
pwcorr estudo lnv4718, sig
```

Anos de estudo estão fortemente correlacionados com o logaritmo do rendimento no trabalho principal, com significância estatística ($p < 0,0001$).

Ao elevar ao quadrado a correlação e multiplicar por 100, podemos determinar a porcentagem de variabilidade conjugada entre as variáveis:

```
di (0.4282^2)*100
```

O resultado indica que a variável de anos de estudo tem uma variação positiva de 18,34% com o logaritmo de rendimento no trabalho principal.